Illustrating ordinary amplitude modulation power efficiency

Initialization Code (optional)

Manipulate

```
Manipulate[
 am[μ, fc, fm, showCarrier, showMessage],

 Style["modulation parameters", Bold],
 {{μ, 0.8, "μ"}, 0, 1.3, .1, Appearance → "Labeled", ImageMargins → -2, ImageSize → Small},
  {{fc, 7000, "f_c (hz)"}, 5000, 10 000, 100, Appearance → "Labeled", ImageMargins → -2, ImageSize → Small},
  {{fm, 500, "f_m (hz)"}, 300, 1000, 10, Appearance → "Labeled", ImageMargins → -2, ImageSize → Small},
 Delimiter,
 Style["modulated carrier plot options", Bold],
 {{showCarrier, True, "show carrier "}, {True, False}},
 {{showMessage, True, "show message "}, {True, False}}
 ,
 ControlPlacement → Left,
 Initialization :→
  (

   maxCarrierFreq = 10 000;
   maxMessageFreq = 1000;
   tmax = 3;
   fontSizeForTitles = 14;
   fontSizeForSubTitles = 11;

   am[μ_, fc_, fm_, showCarrier_, showMessage_] :=
    Module[{plotStyleForCarrier, messagePlot, envelopePlot, carrierPlot, p},

     carrierPlot = (1 + μ Cos[2 Pi fm t / 1000]) Cos[2 Pi fc t / 1000];
     messagePlot = Cos[2 Pi fm t / 1000];
     envelopePlot = 1 + μ Cos[2 Pi fm t / 1000];

     If[showCarrier == True && showMessage == True, {plotStyleForCarrier = {Red, Blue, {Black, Dashed, Thin}},
       p = {carrierPlot, messagePlot, envelopePlot}}];

     If[showCarrier == True && showMessage == False, {plotStyleForCarrier = {Red, {Black, Dashed, Thin}},
       p = {carrierPlot, envelopePlot}}];

     If[showCarrier == False && showMessage == True, {plotStyleForCarrier = {Blue, {Black, Dashed, Thin}},
       p = {messagePlot, envelopePlot}}];

     If[showCarrier == False && showMessage == False, {plotStyleForCarrier = {{Black, Dashed, Thin}},
       p = envelopePlot}];


     GraphicsGrid[{
       { Plot[ Cos[2 Pi fc t / 1000] , {t, 0, 2}, TicksStyle → Small,
         Ticks → {{0, 1, 2}, {-1, 0, 1}}, AxesLabel → {Style[Row[{Style["t", Italic], " (ms)"}], Larger]},
         PlotLabel → Style["carrier signal", fontSizeForTitles], PlotStyle → Red, Exclusions → {1}],
```

```
    Plot[ Cos[2 Pi fm t / 1000] , {t, 0, 2}, TicksStyle → Small,
     Ticks → {{0, 1, 2}, {-1, 0, 1}}, AxesLabel → {Style[Row[{Style["t", Italic], " (ms)"}], Larger]},
     PlotLabel → Style["message signal", fontSizeForTitles], PlotStyle → Blue, Exclusions → {1}]
   },

   {Plot[p,
     {t, 0, tmax },
     PlotRange → {All, {-2.5, 2.5}},
     TicksStyle → Small, AxesLabel → {Style[Row[{Style["t", Italic], " (ms)"}], Larger]},
     PlotLabel → Column[{Style["modulated carrier", fontSizeForTitles],
        Style["with envelope and orginal message superimposed", fontSizeForSubTitles]}, Center],
     PlotStyle -> plotStyleForCarrier, ImagePadding → Full, Exclusions → {1}], SpanFromLeft
   },

   {frequencyPlot[fm, fc, μ], SpanFromLeft}},
  Spacings → {Scaled[.1], Scaled[.1]},
  Frame → {False, All}, Dividers → None, AspectRatio → Full, ImageSize → 320]
];
frequencyPlot[fm_, fc_, μ_] := Module[{carrierRight, carrierLeft, upperRight, upperleft,
   lowerRight, lowerleft, carrierPower, sidePower, z = 0.7 * maxCarrierFreq, g, plotLabel},
  carrierRight = {Red, Arrow[{{fc, 0}, {fc, .5 * fc}}] };
  carrierLeft = {Red, Arrow[{{-fc, 0}, {-fc, 0.5 * fc}}]};
  upperRight = {Blue, Arrow[{{fc + fm, 0}, {fc + fm, fc * μ / 4.}}]};
  upperleft = {Blue, Arrow[{{fc - fm, 0}, {fc - fm, fc * μ / 4.}}]};
  lowerRight = {Blue, Arrow[{{-fc + fm, 0}, {-fc + fm, fc * μ / 4.}}]};
  lowerleft = {Blue, Arrow[{{-fc - fm, 0}, {-fc - fm, fc * μ / 4.}}]};
  carrierPower = 0.5;
  sidePower = μ^2 / 4.;

  g = {  Line[{{-maxCarrierFreq - 0.1 * maxCarrierFreq, 0}, {maxCarrierFreq + 0.1 * maxCarrierFreq, 0}}],
    Line[{{0, 0}, {0, .25 * z}}],
    carrierRight,
    carrierLeft,
    upperRight,
    upperleft,
    lowerRight,
    lowerleft,
    Text[Style["-fc", Small], {-fc, -0.1 * z}],
    Text[Style["fc", Small], {fc, -0.1 * z}],
    Text[Style["0", Small], {0, -0.1 * z}]
   };

  plotLabel = Style[Column[{ Style["modulated carrier frequency (magnitude)", fontSizeForTitles] ,
      Row[{Style["carrier power = 0.5 watt, ", fontSizeForSubTitles],
        Style["sidebands power = " <> ToString[NumberForm[sidePower, {3, 3}]] <> " watt",
         fontSizeForSubTitles]}], Style["modulation efficiency = " <>
        ToString[NumberForm[μ^2 / (μ^2 + 2) * 100, {3, 2}]] <> "%", fontSizeForSubTitles]
    }, Center]] ;

  ListPlot[{{0, 0}}, PlotRange → {{-maxCarrierFreq - 0.3 * maxCarrierFreq,
      maxCarrierFreq + 0.3 * maxCarrierFreq}, {-10, 0.5 * maxCarrierFreq}},
   PlotLabel → plotLabel, Ticks → {{-fc, 0, fc}, None},
   AxesLabel → {Style[Row[{Style["f", Italic], " (hz)"}], Larger]}, ImagePadding → Full,
   Epilog → g]
]
```

)

]

modulation parameters

μ 1.2

f_c (hz) 9300

f_m (hz) 1000

modulated carrier plot options

show carrier ☑

show message ☑



Caption

This demonstration illustrates the effect of changing the modulation factor μ (also called the modulation depth) on the power efficiency of ordinary amplitude modulation (AM) technique. This demonstration also shows that a modulation factor μ greater than unity will result in envelope distortion causing difficulities during demodulation. Hence μ must be kept less than unity.

At 100% modulation (μ=1), the power efficiency of AM is 33.33%. This means that at μ=1, the total power carried by the sidebands is 1/3 of the total power. The modulated carrier $s(t)$, shown in the middle graph in this demonstration is defined as $s(t) = (1 + \mu\, m(t)) \cos(2\pi f_c t)$ where $m(t)$ is the message signal defined as $m(t) = \cos(2t\pi f_m)$ and μ is the modulation factor. For simplicity, the amplitude of the carrier signal and the amplitude of the message signal are both kept as unity.

- ☐ Drag Locators
- ☐ Create and Delete Locators
- ☐ Slider Zoom
- ☐ Gamepad Controls
- ☐ Automatic Animation
- ☐ Bookmark Animation

Search Terms (optional)

Amplitude Modulation

Related Links (optional)

http://scienceworld.wolfram.com/physics/AmplitudeModulation.html
http://demonstrations.wolfram.com/AmplitudeModulation/
http://demonstrations.wolfram.com/ModulacionesAMSpanish/

Authoring Information

Contributed by: Nasser M. Abbasi